



*Использование запросов в системе
"1С:Предприятие 8"*

*Методические материалы
для слушателя сертифицированного курса*

Июль, 2014

ПРАВО ТИРАЖИРОВАНИЯ И РАСПРОСТРАНЕНИЯ
МЕТОДИЧЕСКИХ МАТЕРИАЛОВ
ПРИНАДЛЕЖИТ ФИРМЕ "1С"

Получив настоящие материалы для обучения, Вы тем самым даете согласие
не допускать их копирования без письменного
разрешения фирмы "1С"

© ООО "1С", 2014 г.

Фирма "1С", Москва, 123056, а/я 64
Отдел продаж: ул. Селезневская, д.21,
телефон: (495)737-92-57,
факс: (495) 681-44-07,
e-mail: 1c@1c.ru,
URL: <http://www.1c.ru>

Автор материалов: ООО "1С-Учебный центр №3",
(499) 253-58-38, uc3@1c.ru, www.1c-uc3.ru

07-14

Предложения по совершенствованию методических материалов
просьба направлять в отдел организации обучения фирмы "1С"
E-mail: cso@1c.ru

Содержание

СОГЛАШЕНИЯ О ТЕРМИНАХ, ОБОЗНАЧЕНИЯХ И ДОПОЛНИТЕЛЬНЫЕ СОГЛАШЕНИЯ	5
ВВЕДЕНИЕ	6
1. ЦЕЛИ И ЗАДАЧИ КУРСА	7
2. ОБЩИЕ ПРИНЦИПЫ МЕХАНИЗМА	8
2.1. Таблицы, поля базы данных	8
2.2. Структура запроса (описание запроса)	10
3. ТАБЛИЦЫ ИНФОРМАЦИОННОЙ БАЗЫ ДАННЫХ	11
3.1. Таблицы констант.....	11
3.1.1. Описание полей выборки, источники в запросе	11
3.2. Таблица справочников	12
3.3. Вложенные таблицы.....	12
3.4. Таблица плана видов характеристик	13
3.5. Конструкции "Различные", Первые №.....	13
3.6. Фильтрация результатов запроса (отборы).....	14
3.7. Таблица документов	15
3.8. Группировки в запросе	15
3.8.1. Условия на значения агрегатных функций	17
3.9. Указание нескольких источников.....	18
3.9.1. Переход в другую таблицу по точке	18
3.9.2. С помощью конструкции "Где"	19
3.9.3. Использование соединений.....	19
3.10. Упорядочивание результатов запроса	22
3.11. 4.11. Встроенные функции в языке запросов, операция выбора	23
3.11.1. Проверка на NULL.....	25
3.12. Итоги в запросе	25
3.13. Источники-запросы, подзапросы	28
3.13.1. Использование упорядочивания в подзапросах.....	28
3.14. Объединение запросов	29
3.15. Таблица журнала документов	30
3.16. Передача параметров в запрос	30
3.17. Таблицы регистра сведений.....	31
3.17.1. Основная таблица.....	31
3.17.2. Получение актуальных данных	31
3.17.3. Таблица получения среза последних	32
3.17.4. Таблица получения среза первых	32
3.18. Таблицы регистра накопления	33
3.18.1. Основная таблица.....	33
3.18.2. Таблица получения оборотов.....	33
3.18.3. Таблица получения остатков	35
3.18.4. Таблица получения остатков и оборотов.....	35
3.19. Особенности использования параметров виртуальных таблиц	36
3.19.1. Условие как параметр.....	36
3.19.2. Периодичность	38
3.20. Использование временных таблиц, пакетные запросы	38
3.21. Таблицы регистра бухгалтерии (с поддержкой корреспонденции)	40

3.21.1. Основная таблица.....	40
3.21.2. Значения субконто	40
3.21.3. Движения с субконто	41
3.21.4. Остатки.....	41
3.21.5. Обороты	41
3.21.6. Остатки и обороты	42
3.21.7. Обороты Дт Кт	42
3.22. Таблицы последовательностей	43
3.22.1. Основная таблица.....	43
3.22.2. Таблица границ.....	43
3.23. Таблицы, используемые для решения расчетных задач.....	43
3.24. Таблицы регистрации изменений.....	45
3.25. Таблицы внешних источников.....	45
3.26. Особенности построения запросов при ограничении доступа к данным	45
3.27. Особенности написания запросов для динамических списков	46
3.28. Работа с конструктором запроса.	47
4. РАБОТА С ОБЪЕКТОМ "ЗАПРОС"	53
4.1. Способы обхода результатов запроса.....	53
4.2. Конструктор с обработкой результата.....	57
4.3. Вызов конструктора запроса из тонкого клиента	57
4.4. Использование объектной модели запроса.....	58
4.5. Работа с вложенными таблицами	61
4.6. Использование менеджера временных таблиц	62
4.7. Использование предопределенных данных конфигурации	65
4.8. Возможности конструкции "В"	67
4.9. Получение всех результатов пакетного запроса	68
5. ТИПОВЫЕ ОШИБКИ	70
6. РЕКОМЕНДАЦИЯ "НАПОСЛЕДОК"	71

Соглашения о терминах, обозначениях и дополнительные соглашения

Названия диалоговых кнопок, закладок диалоговых панелей, названия пунктов меню, имена других объектов, будут даваться в двойных кавычках, например, "ОК", "Услуги", "Предприятие", "Контрагент" и т.д.

Значения и типы данных будут даваться в угловых скобках: <дата>, <СправочникСсылка.Организации>.

Обращение к пункту меню будет даваться в последовательном перечислении родительских пунктов через значок стрелки вправо "/", например, "Конфигурация/ Поддержка/ Обновить конфигурацию".

Практические задания делятся на Практикумы (задания, выполняемые самостоятельно), упражнения (выполняемые вместе с преподавателем). Практические задания оформляются:

Практикум № _____

Ссылки на процедуры и функции в основном тексте будут даваться в двойных кавычках "" без скобок.

Важные дополнения к материалу даются:

Важно! _____

Введение

"1С:Предприятие" является универсальной системой автоматизации учетной деятельности организаций. За счет своей универсальности система "1С:Предприятие" может быть использована для автоматизации самых различных участков деятельности организаций, предприятий.

Основной особенностью системы "1С:Предприятие" является ее конфигурируемость. Собственно платформа "1С:Предприятие" представляет собой совокупность механизмов, предназначенных для манипулирования различными типами объектов предметной области. Конкретный набор объектов, структуры информационных массивов, алгоритмы обработки информации определяет конкретная конфигурация. Вместе с конфигурацией система "1С:Предприятие" выступает в качестве уже готового к использованию программного продукта, ориентированного на определенные типы предприятий и классы решаемых задач.

Функционирование системы делится на два процесса:

- конфигурирование (описание модели предметной области средствами, предоставляемыми системой)
- исполнение (обработка данных предметной области).

Процесс конфигурирования в свою очередь распадается на несколько составляющих:

- "визуальное" создание структуры конфигурации (таких объектов конфигурации как справочники, документы и т.п.)
- настройка диалоговых форм объектов
- определение специфики поведения объектов, форм (прописывание кода на языке системы в определенных местах конфигурации)
- создание интерфейсов, наборов прав пользователей (ролей)

В процессе исполнения система уже оперирует конкретными понятиями, описанными на этапе конфигурирования (справочниками товаров и организаций, накладными и т.д.). Обработка информации, введенной пользователями, выполняется как штатными средствами системы, так и с использованием алгоритмов, созданных на этапе конфигурирования.

Можно сказать, что система 1С:Предприятие 8 предоставляет две модели доступа к данным:

- Объектная модель (доступ идет через свойства, методы объектов)
- Табличная модель (конфигурация представляется набором таблиц).

В рамках данного тренинга основной упор будет произведен на табличную модель доступа к данным (на работу с запросами).

1. Цели и задачи курса

Данный курс рассчитан на слушателей, имеющих опыт работы с предметно-ориентированной системой "1С:Предприятие 8": имеющих представление о функциональности объектов системы, особенностях настройки объектов конфигурации, обладающих опытом работы на встроенном языке программного комплекса.

Курс является "продолжением" базового курса "Введение в конфигурирование. Основные объекты".

В процессе обучения Вы ознакомитесь с особенностями языка запросов системы, использования самого объекта "Запрос", порядком работы с результатом запроса, выборкой из результата запроса и т.п.

Курс имеет выраженную практическую направленность, и в ходе его прохождения Вы приобретете практические навыки по написанию запросов как "вручную", так и с использованием конструктора запросов.

Считается, что после ознакомления с материалом данного курса вы будете обладать знаниями и практическими навыками необходимыми для написания запросов любой сложности.

2. Общие принципы механизма

В любой системе если речь заходит о запросах, то упоминаются следующие понятия:

- Источники данных для запроса (таблицы системы, состав полей таблиц)
- Структура запроса, язык написания запроса
- Порядок работы с запросом, получение данных, обход полученного результата.

Рассмотрим эти понятия несколько подробнее

2.1. Таблицы, поля базы данных

Таблицы в "1С:Предприятии 8" подразделяются на два основных класса: реальные и виртуальные.

Для простоты на первых порах можно считать, что:

- Реальные таблицы хранятся в базе данных. В случае использования реальной таблицы могут присутствовать вычисляемые поля, значения которых вычисляются как функция нескольких разных полей. Отдельный подкласс таблиц образуют так называемые объектные таблицы. Эти таблицы предназначены для хранения состояния объектов системы, таких как справочники, документы и т.д. В таких таблицах присутствует поле "Ссылка" (ссылка на объект, данные которого содержит текущая запись таблицы).
- Виртуальные таблицы в базе данных не хранятся. При обращении к информации виртуальных таблиц система автоматически собирает информацию реальных таблиц для выполнения запроса. Виртуальная таблица может быть параметризована.

В свою очередь любая таблица состоит из набора полей. В качестве поля таблицы может фигурировать:

- "Обычное поле" (содержащее какое-либо значение, либо значение типа "Null")
- вложенная таблица

Основное отличие обычного поля от вложенной таблицы состоит в том, что в рамках одной записи обычному полю соответствует одно единственное значение, а вложенной таблице соответствует значение типа "РезультатЗапроса" с заранее заданным набором колонок.

Можно проиллюстрировать данное понятие следующим образом:

Дата	Продавец	Товары
01.09.2003 15:00:10	Продавцов П.П.	
08.09.2003 17:10:16	Продавайлов В.В.	

Товар	Количество
Ручка	10
Карандаш	50

Товар	Количество
Метла	5

Если продолжить разговор о полях, содержащих какие-либо значения, можно отметить: поле может содержать значение одного типа, может содержать значения нескольких типов (иметь составной тип), при этом для конкретной записи поле содержит значение "одного" типа.

В качестве "простых" типов могут выступать:

- Число (десятичное число)
- Строка (строка фиксированной, или неограниченной длины)
- Дата (дата, время)
- Булево (истина или ложь)
- Неопределенно
- Null

Помимо этих типов в качестве типов полей могут выступать: хранилище значения, момент времени, типы, появившиеся после определения в конфигураторе объектов конфигурации...

Важно!

На самом деле нужно осознавать, что даже понятие "реальная таблица" не тождественно понятию "физическая таблица" (это, к примеру, можно проиллюстрировать понятием "вложенная таблица"). Разработчик прикладного решения не имеет возможности работать с таблицами базы данных (физическими таблицами). Язык запросов работает по "логическим" таблицам (они и подразделяются на реальные и виртуальные).

Эти логические таблицы обладают неким "дополнительным" собственным поведением. Т.е. когда разработчик запросом обращается к логическим таблицам, система при обращении к физическим таблицам дополняет запрос разработчика собственным подзапросом, дополнительными полями.

С этой точки зрения реальная таблица является, как правило, "прямым отражением" данных физической таблицы. Подзапрос, добавляемый системой, довольно прост (получение представления, вложенная таблица).

Виртуальная таблица обладает более сложным "поведением". Т.е. при обращении к одной и той же виртуальной таблице с разными значениями параметров платформа может добавлять различные подзапросы (в определенных случаях даже несколько). При этом данные возвращаемые запросом, могут

подвергаться дополнительной обработке (быть сгруппированными, упорядоченными)

2.2. Структура запроса (описание запроса)

В качестве основного механизма получения данных в 1С:Предприятие 8 используется механизм запросов. Запросы получают необходимые данные из таблиц информационной базы (а начиная с определенного релиза и из внешних). Для того чтобы запрос получил "нужные" данные их состав необходимо "указать" (поставить задачу запросу на выборку необходимых данных, отвечающих заданным критериям). "Постановка задачи" производится на специальном языке запросов, поддерживаемым программным комплексом 1С:Предприятие 8. С помощью конструкций данного языка составляется текст запроса.

Можно сказать, что текст запроса – это инструкция, в соответствии с которой должен быть выполнен запрос. Данный текст имеет определенную структуру.

Возможная структура текста запроса приводится ниже:

- Описание запроса
 - Выбрать [Различные] [Первые <Количество>]
 - <Список полей выборки>
 - [Из <Список источников>]
 - [Где <Условие отбора>]
 - [Сгруппировать По <Поля группировки>]
 - [Имеющие <Условия отбора>]
 - [Для изменения [[Of]<Список таблиц верхнего уровня>]]
- Объединение запросов
- Упорядочивание результатов
- Автоупорядочивание
- Описание итогов

В минимальном варианте текст запроса может содержать только конструкцию "Выбрать <Поля выборки>".

После составления текста запроса, его необходимо передать специальному объекту "запрос" и затем выполнить метод данного объекта, результатом выполнения будет объект "результат запроса", который необходимо "обойти".

3. Таблицы информационной базы данных

Рассматривая более подробно таблицы – источники данных для запроса, будем одновременно знакомиться с языком описания запросов. В этом поможет обработка "Знакомство с запросом".

Следует отметить, что структура таблиц приведена в разделе "Работа с запросами/ Таблицы запросов" синтакс-помощника.

3.1. Таблицы констант

Для получения данных из констант может использоваться две таблицы: "Константы" и "Константа.ИмяКонстанты". На каждую константу в таблице "Константы" определяется поле с именем, совпадающим с именем константы, задаваемым в конфигураторе.

3.1.1. Описание полей выборки, источники в запросе

Для получения значения одной константы можно использовать следующий запрос:

```
ВЫБРАТЬ
    Константа.Организация.Значение
```

Просмотреть значения нескольких констант можно с помощью простого запроса:

```
Выбрать
Константы.Адрес,
Константы.ДатаБлокировки
```

После обязательного ключевого слова "Выбрать" следует описание полей выборки. Описание полей разделяются запятыми. После последнего описанного поля запятая не ставится.

Другой вариант написания предыдущего запроса:

```
Выбрать
Адрес,ДатаБлокировки
Из
Константы
```

Источник запроса (под источником запроса обычно понимается таблица, реальная или виртуальная) указывается после ключевого слова "Из". В общем случае источников может быть несколько. Описание источников разделяется запятыми, после последнего источника запятая не ставится.


Следующий запрос показывает содержимое всех полей в таблице констант

```
Выбрать * Из Константы
```

3.2. Таблица справочников

На каждый справочник, определенный в конфигураторе существует таблица. Имя таблицы складывается из ключевого слова "Справочник" и через точку добавляется имя справочника, заданное в конфигураторе. Например: "Справочник.Номенклатура".

Таблица справочника обладает следующим составом полей:



Важно!

Структура конкретных таблиц зависит от настроек соответствующего объекта конфигурации

В каждой объектной таблице присутствует еще поле "ВерсияДанных", не указанное в синтакс-помощнике.

Практикум № 1

Просмотрите состав полей справочников: "Сотрудники", "Контакты", "Подразделения". Обратите внимание на некоторое "несоответствие" приведенным полям

3.3. Вложенные таблицы

При просмотре состава полей таблицы справочника "Сотрудники" в колонке "Родственники" в каждом поле показывается, что в нем содержится тип "Таблица значений", на самом деле в данном поле находится значение, имеющее тип "Результат запроса".

Для простоты понимания не будем пока оперировать понятием "вложенная" таблица, а будем считать, что на каждую табличную часть существует своя таблица. Имя данной таблицы складывается из трех частей (разделенных точками): ключевое слово "Справочник", "Имени справочника", "Имени табличной части" (следует понимать, что в данном случае речь идет о табличных частях справочника, при работе с другими объектами имя таблицы будет

отличаться). Следующий запрос позволит увидеть наполнение и состав полей табличной части "Родственники" справочника "Сотрудники".

Выбрать *


Из

Справочник.Сотрудники.Родственники

3.4. Таблица плана видов характеристик

Данная таблица мало чем отличается от таблицы справочника (отсутствует поле "Владелец")

Состав полей таблицы:



Практикум № 2

Напишите текст запроса, который позволит просмотреть состав полей и наполнение таблиц определенных в конфигурации планов видов характеристик.

Практикум № 3

Напишите текст запроса, который выведет перечень основных единиц измерения, используемых в записях справочника "Номенклатура".

3.5. Конструкции "Различные", Первые №.

При выполнении предыдущего практикума в результате запроса было видно, что перечень основных единиц измерения повторяется. Для получения в результате запроса необходимого количества "первых" записей или чтобы в результате запроса были только "различные" записи можно использовать "дополнения" к ключевому слову "Выбрать"

- Выбрать Первые №(количество записей)
- Выбрать Различные
- Выбрать Разрешенные (будет рассматриваться позже)

Практикум № 4

Опробуйте на практике вышерассмотренные возможности (конструкции "Первые" и "Различные").

Попробуйте объяснить результат применения конструкции "Различные" (запрос по основным единицам измерения справочника номенклатуры должен выдавать два "пустых" значения)

3.6. Фильтрация результатов запроса (отборы)

Для указания условия отбора используется структура, определяемая ключевым словом "Где <Условие>". Условие, определяемое после данного ключевого слова, действует на момент получения данных запросом.

В условии может использоваться тот факт, что некоторые поля могут содержать значение логического типа. Например:

Выбрать
Код,Наименование
Из Справочник.Номенклатура
Где ЭтоГруппа

Можно было бы написать "ЭтоГруппа=Истина". В условиях можно использовать стандартные логические операции, например

"ЭтоГруппа=Ложь И (Не ПометкаУдаления)"

В условиях (это не обязательно может быть конструкция "Где") помимо обычных операций сравнения могут использоваться "В", "В Иерархии" "Между И", "Подобно", "Есть Null", "Ссылка".

Практикум № 5

Напишите текст запроса, который при выполнении выдаст номенклатурные позиции справочника, у которых в поле основной единицы измерения находится значение Null.

Оператор проверки ссылочного значения "Ссылка" позволяет проверить, является ли значение выражения, указанного справа от него ссылкой на таблицу, указанную слева.

Выбрать *
Из РегистрСведений.ЗначенияСвойств
Где Значение Ссылка Справочник.СвойстваНоменклатуры

В операторе "Подобно" символ <%> - означает любую последовательность символов, <_> один символ, <[...]> один символ из указанного в скобках диапазона, <[^...]> один символ не входящий в указанный в скобках диапазон. Например, шаблон "_a[0-9]%" означает, что в строке первый символ любой, дальше буква "a", затем цифра и подстрока любой длины.


Практикум № 6

Напишите текст запроса, который при выполнении выдаст номенклатурные позиции справочника, наименование которых состоит из трех слов.

3.7. Таблица документов

Каждому документу, определенному в конфигураторе соответствует таблица с именем: состоящим из ключевого слова "Документ" и имени документа, определенным в конфигураторе.

Данная таблица имеет следующий состав полей:



Практикум № 7

- 1. Напишите текст запроса, позволяющий посмотреть поля таблицы документа "ПоступлениеТоваров". Найдите некоторое "несоответствие" с таблицей, представленной в методике.*
 - 2. Напишите текст запроса, позволяющий получить данные из табличной части документа "Товары".*
-

3.8. Группировки в запросе

При выполнении пункта №2 практикума №7 записи по закупленным товарам выводились без какой-либо их обработки (сколько раз номенклатурная позиция встречалась в табличной части документов, столько строк в результате запроса). Если потребуется получить ответ на вопрос: "Сколько, какого товара закупалось", то в этом случае в тексте запроса необходимо использовать группировки.

Данные в запросе могут быть сгруппированы с помощью агрегатных функций, указанных в качестве полей выборки.

Выбрать
Номенклатура Как Товар,
Сумма(Количество)
Из
Документ.ПоступлениеТоваров.Товары
Сгруппировать По Номенклатура

Очень важно помнить, что в большинстве случаев все поля выборки запроса должны делиться на агрегатные функции и поля, по которым ведется группировка и поля вложенных таблиц (по отношению к группируемым полям).

Следующий текст запроса выполниться без ошибки

Выбрать
Номенклатура Как Товар,
Номенклатура.ЦенаПродажи,
Сумма(Количество)
Из
Документ.ПоступлениеТоваров.Товары
Сгруппировать По Номенклатура

А попытка выполнить следующий запрос приведет к ошибке:

Выбрать
Номенклатура Как Товар,
Номенклатура.Код,
Сумма(Количество)
Из
Документ.ПоступлениеТоваров.Товары
Сгруппировать По Номенклатура.Код

При указании группировки псевдоним поля указывать нельзя.

В качестве агрегатных функций можно использовать:

- Сумма (Выражение)
- Среднее (Выражение)
- Минимум (Выражение)

- Максимум (Выражение)
- Количество ([Различные] Выражение)

Агрегатные функции можно использовать и без определения группировок.

Выбрать

Количество(ОснЕдиницаИзмерения) Как КоличествоЗначений,

Количество(*) Как Всего

Из

Справочник.Номенклатура

Практикум № 8

Напишите текст запроса, выполнение которого привело бы к получению данных о суммовых оборотах по поставщикам (какой контрагент, на какую сумму поставил товара).

3.8.1. Условия на значения агрегатных функций

Рассмотрим следующий текст запроса:

Выбрать

Номенклатура Как Товар,

Сумма(Количество), Сумма(Сумма)

Из

Документ.ПоступлениеТоваров.Товары

Сгруппировать По Номенклатура

Имеющие Сумма(Сумма)>1000

Важно!

Следует помнить, что в конструкции "Имеющие" необходимо описывать условия именно на агрегатные функции.

Практикум № 9

Напишите текст запроса, который выдавал бы список покупателей и документов, где количество строк в табличной части было бы меньше 3.

3.9. Указание нескольких источников

Не всегда данные, необходимые для выборки, находятся в одной таблице. Иногда приходится (явным или не явным образом) в качестве источников запроса указывать несколько таблиц.

Попытаемся получить перечень номенклатурных позиций и используемых базовых единиц измерения.

Выбрать

Номен.Наименование, ЕдИзм.База

Из Справочник.Номенклатура Как Номен,

Справочник.ЕдиницыИзмерения Как ЕдИзм

Результат подобного запроса состоит из всех возможных комбинаций записей обеих таблиц. Такой результат мало кого устроит, но так будет всегда, если не указывать способ связи таблиц.

Для решения поставленной задачи можно использовать три различных подхода.

3.9.1. Переход в другую таблицу по точке

Если какое поле таблицы (№1) представляет собой ссылку на запись из другой таблицы (№2), то можно используя конструкцию: Таблица1.ИмяПоля.ИмяПоляТаблицы2 "получить" данные из другой таблицы

Выбрать

Наименование, ОснЕдиницаИзмерения.База

Из Справочник.Номенклатура

В качестве еще одного примера

Выбрать

Наименование, ОснЕдиницаИзмерения.База

Из Справочник.Номенклатура

Где

Не (ОснЕдиницаИзмерения.База Есть Null)

Важно!

Следует понимать, что при обращении к полю другой таблицы "через точку" в запросе, выполняемом к физическим таблицам, реально производится соединение с другой таблицей. Исходя из этого, не следует без необходимости ("на всякий случай") выполнять подобные обращения.

Следует избегать конструкций вида

ГДЕ ОснЕдиницаИзмерения.Ссылка=...

3.9.2. С помощью конструкции "Где"

Выбрать

Номен.Наименование, База

Из Справочник.Номенклатура Номен, Справочник.ЕдиницыИзмерения ЕДИЗМ

Где Номен.ОснЕдиницаИзмерения=ЕДИЗМ.Ссылка

В данном случае хочется отметить, что если есть возможность "Связать по ссылкам", то "связывание" по кодам, наименованиям и т.п. будет не оптимальным (и в общем случае не верным).

3.9.3. Использование соединений

Другим способом указания взаимосвязи таблиц является использование "Соединений". Соединения бывают нескольких видов:

- Внутреннее соединение
- Левое внешнее соединение
- Правое внешнее соединение
- Полное внешнее соединение

В любом случае, когда речь заходит о соединении, существует несколько связанных с этим понятий: Таблица №1, Таблица №2, соединение (его вид и условие соединения).

Рассмотрим эти варианты на следующем примере:

Самолеты	
Наименование	Номер
ИЛ86	АА1
Ту 154	АА2
ИЛ 62	АА3

Пассажиры	
НомерСамолета	ФИО
АА1	Первов
АА2	Второв
АА4	Третий
АА1	Четвертый

Условием соединения будет:

Самолеты.Номер=Пассажиры.НомерСамолета

В результат запроса должны попасть данные:

Наименование самолета	Пассажиры

В соответствии с условием можно выделить записи, для которых условие выполняется:

ИЛ 86	AA1
Ту 154	AA2

AA1	Первов
AA1	Четвертый
AA2	Второв

Записи, неудовлетворяющие условию соединения:

Из Таблицы №1			
ИЛ 62	AA3	?	Null
Из Таблицы №2			
Null	?	AA4	Третий

Теперь рассмотрим варианты соединения:

Внутреннее соединение: в результате выполнения запроса войдут только данные записей из обеих таблиц, для которых выполняется условие соединения т.е.

ИЛ 86	Первов
ИЛ 86	Четвертый
Ту 154	Второв

Левое внешнее соединение: в результате выполнения запроса войдут данные из записей, для которых выполняется условие соединения и "не вошедшие" из Таблицы №1. Можно сказать, что в результате запроса войдут все данные из Таблицы №1, и для тех записей результата запроса, для которых выполнялось условие соединения в полях, куда помещаются данные из таблицы №2, будут стоять значения, для которых условие не выполняется, будет стоять Null.

ИЛ 86	Первов
ИЛ 86	Четвертый
Ту 154	Второв
ИЛ 62	Null

Правое внешнее соединение обратно левому.

ИЛ 86	Первов
ИЛ 86	Четвертый
Ту 154	Второв
Null	Третий

Полное внешнее соединение. В результат запроса войдут как записи, для которых выполнялось условие соединения, так и записи, полученные из "не вошедших" данных из обеих таблиц.

ИЛ 86	Первов
ИЛ 86	Четвертый
Ту 154	Второв
ИЛ 62	Null
Null	Третий

Теперь после рассмотрения "абстрактного примера" перейдем к примеру более "реальному".

Выбрать

Номен.Наименование, База

Из Справочник.Номенклатура Номен

Левое Внешнее Соединение

Справочник.ЕдиницыИзмерения ЕдИзм

По

Номен.ОснЕдиницаИзмерения=ЕдИзм.Ссылка

Практикум № 10

Необходимо написать текст запроса, выполнение которого привело бы к получению следующих данных: список фамилий, имен, отчеств сотрудников компании и данных об их родственниках.

3.10. Упорядочивание результатов запроса

Если прописать запрос по табличной части "Товара" документа "ПоступлениеТоваров" (даже с использованием группировок), то видно, что "особого" порядка не прослеживается.

Когда необходимо упорядочить записи по определенному правилу, то в данном случае может пригодиться конструкция языка запросов "Упорядочить По"

Выбрать
*
Из
Документ.ПоступлениеТоваров
Упорядочить По Контрагент.Наименование Возр, Номер Убыв

Возможные варианты упорядочивания: "Возр", "Убыв", "Иерархия". В качестве имен полей, по которым производится упорядочивание, можно указывать их псевдонимы.

Обратите внимание на особенность упорядочивания по ссылочным полям:

Выбрать
*
Из
Документ.ПоступлениеТоваров.Товары
Упорядочить По Номенклатура
АВТОУПОРЯДОЧИВАНИЕ

Важно!

Упорядочивание по иерархии возможно только по таблицам с иерархией.

Не имеет смысла упорядочивание по полю "Представление"

При обращении к виртуальным таблицам данные уже могут быть упорядочены. При обращении к реальным таблицам, без использования упорядочивания можно получить данные в ПРОИЗВОЛЬНОМ порядке.

Практикум № 11

Напишите текст запроса, который позволит вывести перечень из пяти самых продаваемых (по количеству) товаров.

3.11. Встроенные функции в языке запросов, операция выбора

Данные функции могут использоваться для преобразования значений в запросах, могут указываться в качестве полей выборки запроса.

- Подстрока (Выражение, Начальный символ, Сколько отрезать)
- Год (выражение) – номер года
- Квартал (выражение) – номер квартала
- Месяц (выражение) – номер месяца
- ДеньГода (выражение) – номер дня в годе
- День (выражение) - номер дня в месяце
- Неделя (выражение) – номер недели в году
- ДеньНедели (выражение) номер дня в неделе
- Час (выражение)
- Минута (выражение)
- Секунда (выражение)
- НачалоПериода (Дата, Период)
- КонецПериода (Дата, Период)
- ДатаВремя(г,м,д,ч,м,с)
- ЕстьNULL(Проверяемое значение, Замена)
- Представление(Значение)
- ПредставлениеСсылки(Значение)

Следует отметить, что при написании текста запроса можно использовать простые арифметические операции (+, -, *, /)

Выбрать
Номенклатура,
Количество-НомерСтроки
Из
Документ.ПоступлениеТоваров.Товары

В ряде случаев необходимо привести исходное значение к какому-либо типу, то для этой цели можно использовать конструкцию

Выразить ("Исходное" Как "Нужно")

Данная конструкция может использоваться для "уменьшения" неявно подключаемых таблиц при соединении по полю, содержащему ссылки на несколько таблиц

Пример:

ВЫБРАТЬ

 ВЫРАЗИТЬ (ОстаткиНоменклатуры.Регистратор КАК
Документ.РеализацияТоваров).Номер,

 ОстаткиНоменклатуры.Номенклатура,

 ОстаткиНоменклатуры.Количество

ИЗ

РегистрНакопления.ОстаткиНоменклатуры КАК ОстаткиНоменклатуры

ГДЕ

ОстаткиНоменклатуры.Регистратор ССЫЛКА Документ.РеализацияТоваров

Практикум № 12

Напишите текст запроса, который позволит вывести контактные данные сотрудников компании (только сотрудников).

Следующий пример показывает возможности конструкции "выбор"

Выбрать

Номенклатура,

Выбор

 Когда Количество>10 Тогда "Больше 10"

 Когда Количество<=10 Тогда "Меньше или равно 10"

Конец Количество

Из Документ.ПоступлениеТоваров.Товары

Допустим следующий вариант:

Выбрать Номенклатура,

Выбор

 Когда Сумма(Количество)>100 Тогда "Больше 10"

 Когда Сумма(Количество)<=100 Тогда "Меньше или равно 10"

Конец Количество

Из

Документ.ПоступлениеТоваров.Товары

Сгруппировать По Номенклатура

3.11.1. Проверка на NULL

В ряде случаев может потребоваться проведение замены значения Null на какое-либо другое значение непосредственно в момент получения данных запросом. Для этой цели можно использовать два подхода:

Первый с использованием конструкции "Выбор"

```
Выбор
    Когда КоличествоОстаток Есть Null Тогда 0
    Иначе КоличествоОстаток
Конец Количество
```

Вторым способом является использование функции ЕстьNull("Проверяемое значение", "Замена").

```
ВЫБРАТЬ
    ЕстьNULL(Номенклатура.ОснЕдиницаИзмерения,"нет")
ИЗ
    Справочник.Номенклатура КАК Номенклатура
```

Обычно второй способ является более предпочтительным (более компактная форма записи, выполняется быстрее и в случае если "Проверяемое значение" имеет тип либо "Число", либо "Строка" производится приведение "Замены" к типу проверяемого значения).

3.12. Итоги в запросе

Для получения итогов в результате запроса в тексте запроса необходимо определить конструкцию "Итоги". Итоги добавляются в результат запроса как итоговые строки.

```
Выбрать
Номенклатура,Цена,Количество,Сумма
Из
Документ.ПоступлениеТоваров.Товары
Итоги Сумма(Количество), Сумма(Сумма) По Номенклатура
```

Посмотрите результат выполнения запроса и сравните его с результатом следующего запроса:

```
Выбрать
Номенклатура,Цена,Количество,Сумма
Из
Документ.ПоступлениеТоваров.Товары
Итоги Сумма(Количество), Сумма(Сумма) По Номенклатура Иерархия
```

Посмотрите результат, если вместо ключевого слова "Иерархия" указать "Только Иерархия".

Следующий запрос показывает, что в качестве итоговых группировок при расчете итогов могут выступать поля и не принадлежащие указанному источнику.

Выбрать

Ссылка.Контрагент,Номенклатура,Цена,Количество,Сумма

Из

Документ.ПоступлениеТоваров.Товары

Итоги Сумма(Количество), Сумма(Сумма)

По Ссылка.Контрагент, Номенклатура

Отдельно следует упомянуть так называемые "Общие" итоги. Общие итоги всегда идут первой строчкой в результате выполнения запроса.

Выбрать

Ссылка.Контрагент,Номенклатура,Цена,Количество,Сумма

Из

Документ.ПоступлениеТоваров.Товары

Итоги Сумма(Количество), Сумма(Сумма)

По Общие,Ссылка.Контрагент,Номенклатура

Допускается использовать в итогах конструкцию "Выбор".

ВЫБРАТЬ

ПоступлениеТоваровТовары.Ссылка.Контрагент КАК Контрагент,

ПоступлениеТоваровТовары.Номенклатура КАК Номенклатура,

СУММА(ПоступлениеТоваровТовары.Количество) КАК Количество,

СУММА(ПоступлениеТоваровТовары.Сумма) КАК Сумма,

СУММА(ПоступлениеТоваровТовары.Ссылка.СуммаДокумента) КАК
СуммаДокумента

ИЗ

Документ.ПоступлениеТоваров.Товары КАК ПоступлениеТоваровТовары

СГРУППИРОВАТЬ ПО

ПоступлениеТоваровТовары.Ссылка.Контрагент,

ПоступлениеТоваровТовары.Номенклатура

ИТОГИ

СУММА(Количество),
СУММА(Сумма),
Выбор Когда Номенклатура Есть Null Тогда
СУММА(СуммаДокумента)
Конец Как СуммаДокумента
ПО
Контрагент, Номенклатура

Кроме этого можно указывать расчетные поля (наряду с агрегатными функциями итогов)

ВЫБРАТЬ

ПоступлениеТоваровТовары.Ссылка.Контрагент КАК Контрагент,
ПоступлениеТоваровТовары.Номенклатура КАК Номенклатура,
СУММА(ПоступлениеТоваровТовары.Количество) КАК Количество,
СУММА(ПоступлениеТоваровТовары.Сумма) КАК Сумма,
СУММА(ПоступлениеТоваровТовары.Ссылка.СуммаДокумента) КАК
СуммаДокумента,
Выразить(СУММА(Сумма)/СУММА(ПоступлениеТоваровТовары.Ссылка.СуммаДокумента) Как Число(4,3)) Как Относительно

ИЗ

Документ.ПоступлениеТоваров.Товары КАК ПоступлениеТоваровТовары

СГРУППИРОВАТЬ ПО

ПоступлениеТоваровТовары.Ссылка.Контрагент,
ПоступлениеТоваровТовары.Номенклатура

ИТОГИ

СУММА(Количество),
СУММА(Сумма),
Выразить(СУММА(Сумма)/СУММА(СуммаДокумента)
Как Число(3,2)) Как Относительно

ПО

Контрагент, Номенклатура

Важно!

При использовании в тексте запроса раздела "Итоги" результат запроса становится больше, за счет появления в нем итоговых записей. Вместе с этими записями в результате запроса и "появляется" иерархия (в этом случае можно говорить о разных вариантах обхода результата запроса).

При определении произвольного выражения в итогах у данного выражения должен определяться псевдоним. Такое поле должно присутствовать в полях выборки.

3.13. Источники-запросы, подзапросы

В списке источников запроса в качестве таблицы – источника может использоваться вложенный запрос. В этом случае описание источника содержит описание вложенного запроса. Рассмотрим пример:

```
Выбрать
Контрагент,Сумма(Сумма)
Из
(
Выбрать
Ссылка.Контрагент,Номенклатура,Сумма(Сумма) Как Сумма
Из
Документ.ПоступлениеТоваров.Товары
Сгруппировать По
Ссылка.Контрагент,Номенклатура
) Как Вложенный
Сгруппировать По
Контрагент
```

Следует отметить, что вложенные запросы можно использовать и в конструкциях "В ()", "В ИЕРАРХИИ ()".

3.13.1. Использование упорядочивания в подзапросах

При описании подзапросов можно использовать возможность упорядочивания (конструкцию "Упорядочить По") в случае, если количество получаемых записей ограничено конструкцией "Первые N". Но в данном случае можно проводить упорядочивание только по полям, которые входят в поля выбираемые запросом.

Практикум № 13

Напишите текст запроса, по результатам которого можно будет увидеть поставщиков пяти самых дорогих (по ценам продажи) номенклатурных позиций.

3.14. Объединение запросов

В языке запросов имеется возможность объединять несколько запросов, при этом записи, полученные с помощью каждого из объединяемых запросов, будут собраны в один результат запроса.

При объединении каждый запрос собирает данные независимо, а такие операции, как упорядочивание и расчет итогов выполняются уже над результатом объединения запросов.

Поля результата запроса будут называться так, как описано в списке полей первого из объединяемых запросов. Поля выборки остальных запросов сопоставляются с полями результата в соответствии с порядком их следования. Объединяемые запросы должны иметь одинаковое количество полей выборки.

Выбрать

Номенклатура, Ссылка как Док, Сумма(Количество)

Из Документ.ПоступлениеТоваров.Товары

Сгруппировать По

Номенклатура, Ссылка

Объединить Все

Выбрать

Номенклатура, Ссылка, Сумма(Количество)

Из

Документ.РеализацияТоваров.Товары


Сгруппировать По

Номенклатура, Ссылка

Упорядочить По Номенклатура

3.15. Таблица журнала документов

На каждый журнал в табличном представлении базы данных определяется соответствующая таблица. Ее имя складывается из ключевого слова "ЖурналДокумента." плюс имя журнала, определенное в конфигураторе.



Для записей документов, реквизиты которых не отражаются в определенной графе, в соответствующем поле будет находиться значение Null.

Обратиться к таблице можно, используя текст запроса:

```
Выбрать * Из ЖурналДокументов.Складские
```

3.16. Передача параметров в запрос

Очень часто встает задача передачи каких-либо значений (параметров) в запрос. К примеру, это могут быть значения условий, накладываемых на запрос. В таком случае в тексте запроса имя параметра прописывается следующим образом:

```
Где Номенклатура=&Номен
```

Практикум № 14


1. *Напишите текст запроса, который при выполнении содержал бы документы только за указанный период. Можно использовать конструкцию "Между &НачПериода и &КонПериода" или используя конструкцию ДатаВремя.*
2. *Напишите текст запроса, выполнение которого приведет к получению остатка по выбранной номенклатурной позиции (использовать таблицы документов, что на самом деле не соответствует концепции системы)*

3.17. Таблицы регистра сведений

3.17.1. Основная таблица

Имя таблицы: "РегистрСведений.ИмяРегистра".

Состав полей:



Практикум № 15

1. Напишите текст запросов, позволяющих просматривать наполнение регистров сведений "Валюты", "Цены поставок"
2. Напишите текст запроса, результатом выполнения которого будет получение "истории" значений выбранной валюты.

3.17.2. Получение актуальных данных

Рассмотрим запрос, возвращающий информацию об актуальных курсах валют.

Выбрать

Срез.Валюта, Рег.Курс, Рег.Кратность

ИЗ

(

ВЫБРАТЬ

МАКСИМУМ(Период) Период, Валюта

ИЗ РегистрСведений.Валюты

СГРУППИРОВАТЬ ПО Валюта

) Срез

ВНУТРЕННЕЕ СОЕДИНЕНИЕ

РегистрСведений.Валюты Рег














ПО Срез.Валюта = Рег.Валюта И Срез.Период = Рег.Период

Получите наиболее актуальные цены номенклатуры по поставщикам (регистр сведений "Цены поставщиков")

3.17.3. Таблица получения среза последних

Виртуальная таблица. Она предназначена для получения наиболее близких в "прошлое" записей регистра сведений на указанную дату (включительно). Включает только активные записи.

Состав полей таблицы:

- ☉  Таблица среза последних
 - ☉  Поля
 -  <Имя измерения>
 -  <Имя общего реквизита>
 -  <Имя реквизита>
 -  <Имя ресурса>
 -  Активность
 -  НомерСтроки
 -  Период
 -  Регистратор
 - ☉  Параметры
 -  Период
 -  Условие


Выбрать

*

Из РегистрСведений. Валюты.СрезПоследних(&НачПериода)

3.17.4. Таблица получения среза первых

Виртуальная таблица. Предназначена для получения наиболее близких в "будущее" записей регистра сведений на указанную дату (включительно). Включает только активные записи.

- ☉  Таблица среза первых
 - ☉  Поля
 -  <Имя измерения>
 -  <Имя общего реквизита>
 -  <Имя реквизита>
 -  <Имя ресурса>
 -  Активность
 -  НомерСтроки
 -  Период
 -  Регистратор
 - ☉  Параметры
 -  Период
 -  Условие

Выбрать

* Из РегистрСведений. Валюты.СрезПервых(&НачПериода)

Практикум № 17

1. Сформируйте текст запроса, выполнение которого приведет к получению данных о курсах валют на выбранную дату.
2. Напишите текст запроса, результат выполнения которого будет получение значения курса выбранной валюты, действующего на выбранную дату. (При этом не надо пользоваться конструкцией "Где" запроса).

3.18. Таблицы регистра накопления**3.18.1. Основная таблица**

Реальная таблица. Предназначена для получения записей регистра накопления. Имя таблицы состоит из ключевого имени "РегистрНакопления." плюс имя регистра, определенное в конфигураторе.

Состав полей таблицы:

- ☉ Таблицы регистра накопления
 - ☉ Основная таблица
 - ☉ Поля
 - ☐ <Имя измерения>
 - ☐ <Имя общего реквизита>
 - ☐ <Имя реквизита>
 - ☐ <Имя ресурса>
 - ☐ Активность
 - ☐ ВидДвижения
 - ☐ МоментВремени
 - ☐ НомерСтроки
 - ☐ Период
 - ☐ Регистратор


Выбрать *

Из РегистрНакопления.ОстаткиНоменклатуры

3.18.2. Таблица получения оборотов

Виртуальная таблица. Предназначена для получения оборотов за период по регистру накопления. При расчете итогов учитываются только активные записи. Данная таблица определена для всех (обоих) видов регистров накопления.

Состав полей таблицы:



Параметры "Начало", "Окончание" задают период, за который получают обороты.

В качестве периодичности может задаваться один из следующих вариантов:

- Период (не разворачивать)
- Запись
- Регистратор
- Секунда
- Минута
- Час
- День
- Неделя
- Декада
- Месяц
- Квартал
- Полугодие
- Год

Посмотреть таблицу можно с помощью запроса:









Выбрать *

Из РегистрНакопления.ОстаткиНоменклатуры.Обороты(„Неделя)

3.18.3. Таблица получения остатков

Виртуальная таблица. Предназначена для получения остатков по регистру накопления. При расчете итогов учитываются только активные записи. Таблица существует только для регистров остатков.

Состав полей таблицы:























- ☉  Таблица остатков
 - ☉  Поля
 -  <Имя измерения>
 -  <Имя общего реквизита>
 -  <Имя ресурса>Остаток
 - ☉  Параметры
 -  Период
 -  Условие

Выбрать *

Из РегистрНакопления.ОстаткиНоменклатуры.Остатки(&НачПериода)

3.18.4. Таблица получения остатков и оборотов

Виртуальная таблица предназначена для получения остатков и оборотов за период по регистру накопления. При расчете итогов учитываются только активные записи. Таблица существует только для регистров остатков.

- ☉  Таблица остатков и оборотов
 - ☉  Поля
 -  <Имя измерения>
 -  <Имя общего реквизита>
 -  <Имя ресурса>КонечныйОстаток
 -  <Имя ресурса>НачальныйОстаток
 -  <Имя ресурса>Оборот
 -  <Имя ресурса>Приход
 -  <Имя ресурса>Расход
 -  НомерСтроки
 -  Период
 -  ПериодГод
 -  ПериодДекада
 -  ПериодДень
 -  ПериодКвартал
 -  ПериодМесяц
 -  ПериодМинута
 -  ПериодНеделя
 -  ПериодПолугодие
 -  ПериодСекунда
 -  ПериодЧас
 -  Регистратор

- ☐ Параметры
 - ☐ Начало периода
 - ☐ Конец периода
 - ☐ Периодичность
 - ☐ Метод дополнения периодов
 - ☐ Условие

Возможные значения метода дополнения:

- Движения (выдаются те периоды, в которых были движения)
- ДвиженияИГраницыПериода (выбираются периоды, по которым были движения и существовали остатки на начало и конец)

Выбрать *

Из РегистрНакопления.ОстаткиНоменклатуры.ОстаткиИОбороты(„Неделя)

Практикум № 18

1. Напишите текст запроса, который при выполнении позволил бы получать остаток выбранной номенклатурной позиции на интересующую дату.
2. Напишите текст запроса, который при выполнении выдавал бы данные, необходимые для "оборотки по месяцам" (номенклатурная позиция, остаток на начало периода, приход, расход, остаток на конец периода) за выбранный интервал дат.

3.19. Особенности использования параметров виртуальных таблиц

3.19.1. Условие как параметр

Использование условия как параметра виртуальных таблиц регистров имеет ряд особенностей (в зависимости от вида регистров).

Рассмотрим два запроса.

Первый:

ВЫБРАТЬ

НоменклатурыОстатки.Номенклатура,

НоменклатурыОстатки.КоличествоОстаток,

НоменклатурыОстатки.СуммаОстаток

ИЗ

РегистрНакопления.ОстаткиНоменклатуры.Остатки(&НачПериода)

КАК НоменклатурыОстатки

ГДЕ НоменклатурыОстатки.Номенклатура = &ЗначениеСсылка

Второй запрос:

ВЫБРАТЬ

НоменклатурыОстатки.Номенклатура,

НоменклатурыОстатки.КоличествоОстаток,

НоменклатурыОстатки.СуммаОстаток

ИЗ

РегистрНакопления.ОстаткиНоменклатуры.Остатки(&НачПериода, Номенклатура = &ЗначениеСсылка) КАК НоменклатурыОстатки

В первом запросе при обращении к виртуальной таблице система выполняет расчет остатков ВСЕХ номенклатурных позиций и только после этого (из полученной таблицы) производится выборка одной номенклатурной позиции. При большом количестве номенклатуры (в остатках) получение остатка одной позиции таким способом будет крайне не оптимальным вариантом (это будет грубой ошибкой программиста).

При выполнении второго запроса остаток будет рассчитываться только на указанную номенклатурную позицию.

Подобная особенность характерна как для виртуальных таблиц регистров накопления, так и для виртуальных таблиц регистров бухгалтерии и расчета.

Использование условий в параметрах виртуальных таблиц регистров сведений может иметь несколько иную особенность.

Рассмотрите следующие два запроса, первый:

ВЫБРАТЬ

ВалютыСрезПоследних.Период,

ВалютыСрезПоследних.Валюта,

ВалютыСрезПоследних.Курс,

ВалютыСрезПоследних.Кратность,

ВалютыСрезПоследних.Источник

ИЗ

РегистрСведений.Валюты.СрезПоследних(&НачПериода, Источник = &ЗначениеСсылка) КАК ВалютыСрезПоследних

Второй:

```
ВЫБРАТЬ
    ВалютыСрезПоследних.Период,
    ВалютыСрезПоследних.Валюта,
    ВалютыСрезПоследних.Курс,
    ВалютыСрезПоследних.Кратность,
    ВалютыСрезПоследних.Источник
ИЗ
    РегистрСведений.Валюты.СрезПоследних(&НачПериода) КАК
    ВалютыСрезПоследних
Где Источник = &ЗначениеСсылка
```

Просмотрите результаты выполнения этих запросов в случае, когда в параметр запроса передается элемент справочника с наименованием "РБК" (справочник "Источники получения данных")

Важно!

При использовании параметра "Условие" совершенно не нужно это условие "дублировать" в разделе "Где" текста запроса. Также следует обратить внимание на тот факт, что данные возвращаются уже сгруппированными.

3.19.2. Периодичность

У виртуальных таблиц тем или иным образом связанных с получением оборотов за период есть параметр, позволяющий определять периодичность получаемых данных.

Следует внимательно относиться к выбору значения этого параметра. При желании получить данные по регистратору нужно, чтобы в этом параметре было выбрано значение "Регистратор" или "Запись". В случае если нужно "добраться" до номера строки следует установить значение "Запись".

3.20. Использование временных таблиц, пакетные запросы

Если в тексте запроса использовать конструкцию "Поместить", то в результате выполнения запроса будет создана временная таблица. В самом результате запроса будет указано количество записей, размещенных в созданной временной таблице.

Пакетный запрос представляет собой возможность в одном тексте (через ";") прописать несколько запросов. Порядок исполнения запросов будет соответствовать порядку их описания.

Рассмотрим пример:

```
Выбрать Различные  
Номенклатура  
Поместить СписокТоваров  
Из  
Документ.РеализацияТоваров.Товары  
Где  
Ссылка=&ЗначениеСсылка  
;  
Выбрать  
Номенклатура,  
КоличествоОстаток  
Из  
РегистрНакопления.ОстаткиНоменклатуры.Остатки(&НачПериода,Номенклатура В  
(Выбрать Номенклатура Из СписокТоваров))
```

Важно!

В тексте запроса, в котором создается временная таблица нельзя использовать конструкцию "Итоги".


Практикум № 19

Напишите текст запроса, выполнение которого приведет к получению позиций справочника номенклатуры, по которым ни разу не оформлялся, ни один документ. Запрос выполнить с применением пакетного запроса (и по таблицам документов "ПоступлениеТоваров", "РеализацияТоваров").

3.21. Таблицы регистра бухгалтерии (с поддержкой корреспонденции)

3.21.1. Основная таблица

Реальная таблица, состав полей следующий.




Просмотреть состав полей можно с помощью запроса:

```
Выбрать * Из РегистрБухгалтерии.Основной
```

3.21.2. Значения субконто

Таблица хранит значения субконто записей регистра бухгалтерии. Связь с таблицей записей регистра бухгалтерии осуществляется через поля: "Регистратор", "Номер строки". Поле "Корреспонденция" показывает, к какой части записи относится данное значение.

Имя таблицы: "РегистрБухгалтерии.ИмяРегистра.Субконто"



Пример запроса:

```
Выбрать * Из РегистрБухгалтерии.Основной.Субконто
```


3.21.3. Движения с субконто

Виртуальная таблица. Позволяет получить информацию о записях регистра вместе со значениями субконто.

Имя таблицы "РегистрБухгалтерии.ИмяРегистра.ДвиженияССубконто", состав полей таблицы следующий:

- ☉ Таблица движений с субконто
 - ☉ Поля
 - ☐ <Имя измерения>
 - ☐ <Имя измерения>Дт
 - ☐ <Имя измерения>Кт
 - ☐ <Имя общего реквизита>
 - ☐ <Имя реквизита>
 - ☐ <Имя ресурса>
 - ☐ <Имя ресурса>Дт
 - ☐ <Имя ресурса>Кт
 - ☐ Активность
 - ☐ ВидСубконтоДт<Номер субконто>
 - ☐ ВидСубконтоКт<Номер субконто>
 - ☐ МоментВремени
 - ☐ НомерСтроки
 - ☐ Период
 - ☐ Регистратор
 - ☐ СубконтоДт<Номер субконто>
 - ☐ СубконтоКт<Номер субконто>
 - ☐ СчетДт
 - ☐ СчетКт
 - ☉ Параметры
 - ☐ Начало периода
 - ☐ Конец периода
 - ☐ Условие
 - ☐ Порядок
 - ☐ Первые

Пример запроса:

Выбрать * Из РегистрБухгалтерии.Основной.ДвиженияССубконто

3.21.4. Остатки

Виртуальная таблица. Позволяет получать остатки на произвольную дату в разрезе счета, измерений и субконто.

Выбрать * Из РегистрБухгалтерии.Основной.Остатки

3.21.5. Обороты

Виртуальная таблица. Позволяет получать обороты на произвольную дату в разрезе счета, корсчета, измерений и субконто и корсубконто.

Выбрать * Из РегистрБухгалтерии.Основной.Обороты

3.21.6. Остатки и обороты

Виртуальная таблица. Позволяет получить обороты и остатки за произвольный период с заданной периодичностью, в разрезе счета, субсчета, измерений.

Выбрать * Из РегистрБухгалтерии.Основной.ОстаткиИОбороты

С недельной периодичностью:

Выбрать

* Из РегистрБухгалтерии.Основной.остаткиИОбороты(„Неделя)

Хотелось бы обратить внимание, что на некоторых счетах может вестись количественный учет (например, на счете учета товаров). В этом случае при проведении расходных документов требуется производить проверку наличия нужного количества товара, определять его себестоимость. Принцип построения таких запросов (получающих вышеописанные данные) аналогичен принципу, используемому при построении запросов по регистрам накопления.

3.21.7. Обороты Дт Кт

Виртуальная таблица. Получает получить обороты между счетами за заданный период, с установленной периодичностью, в разрезе измерений, субконто дебета, субконто кредита.

ВЫБРАТЬ

ОсновнойОборотыДтКт.СчетДт,

ОсновнойОборотыДтКт.СчетКт,

ОсновнойОборотыДтКт.СуммаОборот

ИЗ

РегистрБухгалтерии.Основной.ОборотыДтКт(&НачПериода, &КонПериода, Период, , , ,) КАК ОсновнойОборотыДтКт

Практикум № 20

По таблице оборотов между счетами получите данные необходимые для построения "шахматного" отчета.

СчетДт	СчетКт	СуммаОборот
41	41	
41	60	861 310,00
41	50.1	
60	41	
60	60	
60	50.1	1 000,00
50.1	41	
50.1	60	
50.1	50.1	

Важно!


Хочется отметить, что в реальной жизни для получения подобных отчетов используется функциональность системы компоновки данных.

3.22. Таблицы последовательностей

3.22.1. Основная таблица


Таблица предназначена для получения записей принадлежности документов к последовательности.

Имя таблицы "Последовательность.ИмяПоследовательности", состав полей следующий:



3.22.2. Таблица границ

Данная таблица предназначена для получения границ последовательности. Имя таблицы "Последовательность.ИмяПоследовательности.Границы", состав полей следующий:



3.23. Таблицы, используемые для решения расчетных задач

Для решения расчетных задач в первую очередь используются такие объекты как планы видов расчета и регистры расчета. Сами расчетные задачи решаются исходя из двух видов взаимодействия видов расчета: зависимость по базе и конкуренция за период. Рассмотрим состав таблиц регистров расчета.

Основная таблица содержит записи регистра расчета. Это реальная таблица. Для просмотра состава полей данной таблицы можно использовать запрос

```
ВЫБРАТЬ * ИЗ РегистрРасчета.Основной
```

Таблица "ДанныеГрафика" является виртуальной и предназначена для получения данных по различным периодам в проекции на регистр сведений,

содержащий график работы сотрудников. Данная таблица может использоваться для расчетов видов расчета, конкурирующих за период. Пример запроса:

```
ВЫБРАТЬ *
ИЗ РегистрРасчета.Основной.ДанныеГрафика()
```

При работе с данной таблицей не стоит включать поля "про запас", это может негативно отразиться на времени выполнения запроса.

```
ВЫБРАТЬ
    РасчетДанныеГрафика.ЗначениеФактическийПериодДействия,
    РасчетДанныеГрафика.ДниФактическийПериодДействия
ИЗ
    РегистрРасчета.Основной.ДанныеГрафика
КАК РасчетДанныеГрафика
```

```
ВЫБРАТЬ
    РасчетДанныеГрафика.ЗначениеФактическийПериодДействия,
    РасчетДанныеГрафика.ЗначениеБазовыйПериод
ИЗ
    РегистрРасчета.Основной.ДанныеГрафика
КАК РасчетДанныеГрафика
```

Второй запрос, несмотря на такое же количество виртуальных полей, выполняется медленнее (т.к. идет обращение к разным периодам).

Таблица **Фактический период действия** позволяет просматривать фактические периоды действия видов расчета. В случае если один вид расчета частично был вытеснен другим, то для вытесняемого в данной таблице будет содержаться столько записей, сколько интервалов действия получилось после его вытеснения вытесняющими видами расчета.

```
ВЫБРАТЬ *
ИЗ РегистрРасчета.Основной.ФактическийПериодДействия
```

При работе с видами расчета, взаимодействующими по базе может использоваться таблица:

РегистрРасчета.ИмяРегистра.БазаИмяБазовогоРегистра

Данная таблица является виртуальной и имеет четыре параметра, три из которых являются обязательными (и являются либо массивами, либо списками значения).

В случае если изменились результаты расчета базовых видов расчета, то с помощью таблицы **перерасчетов** можно получить те записи, которые необходимо пересчитать.

ВЫБРАТЬ *
ИЗ РегистрРасчета.Основной.Единственный

Практикум № 21

Напишите текст запроса, который позволит получить данные, необходимые для расчета оклада и сам рассчитанный оклад (вычисляемое поле).

3.24. Таблицы регистрации изменений

С помощью данных таблиц (в случае использования планов обмена) можно просматривать изменения по какому-либо объекту конфигурации, зафиксированные системой. Имя таблицы "ВидОбъекта.ИмяОбъекта.Изменения", состав полей:

- ☉ Таблица изменений справочника
 - ☉ Поля
 - ☐ <Имя общего реквизита>
 - ☐ НомерСообщения
 - ☐ Ссылка
 - ☐ Узел

3.25. Таблицы внешних источников

Предназначена для получения данных из таблиц внешних источников. Имя таблицы: "ВнешнийИсточникДанных.Имя внешнего источника данных.Таблица.Имя таблицы", состав полей:

- ☉ Таблица внешнего источника данных
 - ☉ Поля
 - ☐ <Имя поля>
 - ☐ Представление
 - ☐ Ссылка

3.26. Особенности построения запросов при ограничении доступа к данным

В конфигурации 1С:Предприятие 8 для ряда ролей может быть предусмотрены ограничения доступа к данным. Эти ограничения могут повлиять как на результаты выполнения запросов, так и на их текст.

В методической конфигурации для пользователя "Контрагент" действуют ограничения на работу ("Чтение") с приходными накладными. Он может просматривать документы этого вида только для "своего" контрагента (определенного в справочнике "Пользователи").

Запустите 1С:Предприятие под данным пользователем и попробуйте выполнить следующий запрос:

ВЫБРАТЬ * ИЗ Документ.ПоступлениеТоваров.Товары


При выполнении запроса произойдет ошибка. Правильный вариант следующий:

ВЫБРАТЬ Разрешенные *

ИЗ Документ.ПоступлениеТоваров.Товары

3.27. Особенности написания запросов для динамических списков

В управляемом интерфейсе при определении динамического списка можно использовать произвольный запрос.



Следует обратить внимание на то, что в динамическом списке для произвольного запроса существуют следующие ограничения:

- динамический список не поддерживает работу с пакетными запросами;
- динамический список не поддерживает в запросе объединения, если задана основная таблица;
- динамический список не должен содержать секции УПОРЯДОЧИТЬ ПО, если задана основная таблица. Необходимо использовать запрос без основной таблицы или задавать необходимое упорядочивание через настройки динамического списка;
- в числе полей запроса нельзя использовать поля подзапросов, возвращающих множественное количество значений. Необходимо использовать запрос без основной таблицы;
- запрос не может содержать группировок и агрегатных функций, если задана основная таблица. Необходимо использовать запрос без основной таблицы или задавать необходимые группировки через настройки динамического списка;

- в случае если динамический список отображается в виде иерархического списка или дерева, запрос не должен содержать условий отбора по родителю;
- в случае указания основной таблицы динамического списка запрос не должен содержать инструкций ПЕРВЫЕ и РАЗЛИЧНЫЕ.

Кроме этого, если выбрана основная таблица и используются соединения с другими таблицами, состав записей таблицы не должен меняться (по отношению к тому, если бы динамический список отображал бы только данные этой таблицы). Например, если в указанном примере дополнительно из таблицы остатков вывести в динамический список данные измерения "Склад", то это приведет к ошибкам в работе данного механизма.


3.28. Работа с конструктором запроса.

Умение писать текст запроса "вручную" очень полезно, но в программном комплексе есть возможность строить запросы при помощи конструктора. Вызов конструктора может происходить как в режиме исполнения (но только в обычной форме), так и в режиме конфигурирования. Можно сказать, что каждая из закладок конструктора отвечает за какой-либо раздел текста запроса.


На странице "Таблицы и поля" можно выбрать источники данных запроса, и какие поля этих источников выбираются запросом.

В качестве источника могут использоваться таблицы системы, либо подзапросы, либо временные таблицы. Запустим конструктор (команда контекстного меню поля текстового документа в обработке "Знакомство с запросом").


В качестве таблицы источника выберем таблицу табличной части документа "Реализация товаров". Из этой таблицы выберем поля "Номенклатура" и "Количество" (как показано на рисунке, следующем далее).



Страница "Группировка" "отвечает" за раздел запроса "Сгруппировать По" (определяются поля по которым ведется группировка, определяются виртуальные функции).




На закладке "Условия" определяются условия раздела "Где" или раздела "Имеющие" (зависит от контекста условия).



Флаг "Произвольное" позволяет определять произвольные условия (в том числе использующие подзапросы).


На закладке "Дополнительно" определим, что при выполнении данного запроса будет создаваться временная таблица с именем "ТЧР".




Так как предполагается, то по полю "Номенклатура" будем производить соединение с другой таблицей, то на закладке "Индекс" реализуем следующую настройку:




На закладке "Пакет запросов" создадим следующий запрос пакета. Воспользуемся для этого кнопкой "Добавить" командной панели закладки.




В новом пакете запроса определим источники и выбираемые поля в соответствии с нижеследующим рисунком.




Определим параметры виртуальной таблицы остатков. Для этого необходимо данную таблицу сделать текущей и воспользоваться кнопкой "Параметры виртуальной таблицы".




Параметр "Условие" определим в соответствии с рисунком.



При наличии более одного источника, в конструкторе появляется закладка "Связи" (она "ответственна" за настройку соединений).



Осталось добавить третий запрос в пакет и реализовать в нем инструкцию по удалению временной таблицы, созданной первым запросом пакета.



В результате работы с конструктором был получен запрос следующего вида:

```

ВЫБРАТЬ
    РеализацияТоваровТовары.Номенклатура КАК Номенклатура,
    СУММА(РеализацияТоваровТовары.Количество) КАК Количество
ПОМЕСТИТЬ ТЧР
ИЗ
    Документ.РеализацияТоваров.Товары КАК РеализацияТоваровТовары
ГДЕ
    
```

```
РеализацияТоваровТовары.Ссылка = &ЗначениеСсылка

СГРУППИРОВАТЬ ПО
    РеализацияТоваровТовары.Номенклатура

ИНДЕКСИРОВАТЬ ПО
    Номенклатура
;

////////////////////////////////////

ВЫБРАТЬ
    ТЧР.Номенклатура,
    ТЧР.Количество,
    ОстаткиНоменклатурыОстатки.КоличествоОстаток,
    ОстаткиНоменклатурыОстатки.СуммаОстаток

ИЗ
    ТЧР КАК ТЧР

    ЛЕВОЕ СОЕДИНЕНИЕ
    РегистрНакопления.ОстаткиНоменклатуры.Остатки(
        ,
        Номенклатура В
        (ВЫБРАТЬ
            ТЧР.Номенклатура
            ИЗ
                ТЧР)) КАК ОстаткиНоменклатурыОстатки
    ПО ТЧР.Номенклатура = ОстаткиНоменклатурыОстатки.Номенклатура

////////////////////////////////////

УНИЧТОЖИТЬ ТЧР
```

Практикум № 22

С помощью конструктора запроса реализуйте подобный запрос, но уже по данным бухгалтерского учета.

Практикум № 23

- 1. Напишите текст запроса, выполнение которого приведет к получению данных о том: какой поставщик какой товар закупает.*
 - 2. Получите суммовые обороты (по "недельно") по контактными лицам.*
 - 3. Напишите текст запроса, который при выполнении выдавал бы информацию о суммовых оборотах по месяцам для различных поставщиков. Из результата выполнения запроса должно быть видно: какой общий суммовой оборот есть у поставщика и ниже для каждого поставщика должна идти расшифровка его оборота по месяцам. Месяц прописью*
-

4. Работа с объектом "запрос"

Изучая предыдущие разделы, вы работали с уже готовой обработкой. Пришло время познакомиться с порядком работы с объектом "Запрос" из встроенного языка 1С:Предприятие 8.

В работе из встроенного языка с запросами можно выделить несколько "крупных" этапов:

1. Создание объекта "Запрос"
2. Формирование текста запроса
3. Выполнение запроса
4. Обход результата выполнения запроса

Для более детального ознакомления рассмотрим принципы работы обработки "Знакомство с запросом".

Текст процедуры, определенной в модуле формы приводится ниже:

Процедура КнопкаСформироватьНажатие(Элемент)

```
Запрос=Новый Запрос;  
Запрос.Текст = ЭлементыФормы.ПолеТД.ПолучитьТекст();  
Запрос.УстановитьПараметр("ТЧ", ТЧ.ВыгрузитьКолонку("ТЧСсылка"));  
Запрос.УстановитьПараметр("НачПериода",НачПериода);  
Запрос.УстановитьПараметр("КонПериода",КонПериода);  
Запрос.УстановитьПараметр("ЗначениеСсылка",ЗначениеСсылка);  
Запрос.УстановитьПараметр("ЗначениеЧисло",ЗначениеЧисло);  
Запрос.УстановитьПараметр("ЗначениеБулево",ЗначениеБулево);  
Запрос.УстановитьПараметр("ЗначениеСтрока",ЗначениеСтрока);  
РезультатЗапроса = Запрос.Выполнить();  
ЭлементыФормы.РезультатТП.Значение=РезультатЗапроса.Выгрузить();
```

КонецПроцедуры

4.1. Способы обхода результатов запроса.

До сих пор (в рамках данного курса) не рассматривался вопрос об обходе полученного результата запроса. Он выгружался целиком (хотя следует отметить, что получалась картина равноценная линейному порядку обхода).

Все возможные варианты обхода результата запроса определены в системном перечислении "ОбходРезультатаЗапроса". Значения системного перечисления следующие:

- Прямой
- ПоГруппировкам

- ПоГруппировкамСИерархией

Разбираться с порядком обхода будем на примере обработки "Порядок обхода". Текст запроса, результат которого будет обходить разными вариантами следующий:

```
ВЫБРАТЬ
    ПриходнаяТовары.Ссылка.Контрагент КАК Поставщик,
    ПриходнаяТовары.Номенклатура КАК Номенклатура,
    ПриходнаяТовары.Номенклатура.Наименование КАК
НаименованиеНоменклатуры,
    ПриходнаяТовары.Количество КАК Количество,
    ПриходнаяТовары.Сумма КАК Сумма
ИЗ
    Документ.ПоступлениеТоваров.Товары КАК ПриходнаяТовары
ГДЕ
    ПриходнаяТовары.Ссылка.Дата МЕЖДУ &НачПериода И &КонПериода

УПОРЯДОЧИТЬ ПО
    НаименованиеНоменклатуры
ИТОГИ
    СУММА(Количество),
    СУММА(Сумма)
ПО
    ОБЩИЕ,
    Поставщик,
    Номенклатура ИЕРАРХИЯ
```

Код, процедур, реализующий демонстрируемый механизм приведен ниже:

```
Процедура ВыдатьРекурсивно(ВыборкаДетальная,Знач
Номер,ТабДок,Секция,ОбходРезультата)

    Номер=Номер+1;

    Пока ВыборкаДетальная.Следующий() Цикл
        Секция.Параметры.Поставщик = ВыборкаДетальная.Поставщик;
        Секция.Параметры.Наименование =
        ВыборкаДетальная.НаименованиеНоменклатуры;
```

```
Секция.Параметры.Колво = ВыборкаДетальная.Количество;  
Секция.Параметры.Сумма = ВыборкаДетальная.Сумма;  
Секция.Параметры.ТипЗаписи = ВыборкаДетальная.ТипЗаписи();  
Секция.Параметры.Уровень = ВыборкаДетальная.Уровень();  
Секция.Параметры.Номер = Номер;  
ТабДок.Вывести(Секция, ВыборкаДетальная.Уровень());  
ВыборкаДетальная2 = ВыборкаДетальная.Выбрать(ОбходРезультата);  
ВыдатьРекурсивно(ВыборкаДетальная2,Номер,ТабДок,Секция,ОбходРезультата);  
КонецЦикла;
```

КонецПроцедуры

Процедура ВыполнитьОтчет(ТабДок,ТП,ОбходРезультата) Экспорт

```
Перем ОбщиеИтогиСумма;  
Перем ОбщиеИтогиКоличество;  
  
Макет = ПолучитьМакет("ПродолжениеМакет");  
Секция = Макет.ПолучитьОбласть("Шапка");  
Секция.Параметры.ПериодС = Формат(КонПериода, "ДЛФ=D");  
Секция.Параметры.ПериодПо = Формат(КонПериода, "ДЛФ=D");  
ТабДок.Вывести(Секция);  
Секция = Макет.ПолучитьОбласть("Позиция");  
  
Запрос = Новый Запрос;
```

ТекстЗапроса = "Приведенный текст";

```
Запрос.Текст = ТекстЗапроса;  
Запрос.УстановитьПараметр("НачПериода", НачПериода);  
Запрос.УстановитьПараметр("КонПериода", КонецДня(КонПериода));
```

РезультатЗапроса = Запрос.Выполнить();

```
ТП.Значение=РезультатЗапроса.Выгрузить();

    ТП.СоздатьКолонки();

ТабДок.НачатьАвтогруппировкуСтрок();

Номер=0;

ВыборкаОсновная = РезультатЗапроса.Выбрать(ОбходРезультата);


Пока ВыборкаОсновная.Следующий() Цикл
    Секция.Параметры.Поставщик = ВыборкаОсновная.Поставщик;
    Секция.Параметры.Наименование =
ВыборкаОсновная.НаименованиеНоменклатуры;
    Секция.Параметры.Колво = ВыборкаОсновная.Количество;
    Секция.Параметры.Сумма = ВыборкаОсновная.Сумма;
    Секция.Параметры.Номер = Номер;
    Секция.Параметры.ТипЗаписи = ВыборкаОсновная.ТипЗаписи();
    Секция.Параметры.Уровень = ВыборкаОсновная.Уровень();
    ТабДок.Вывести(Секция, ВыборкаОсновная.Уровень());
    ВыборкаДетальная = ВыборкаОсновная.Выбрать(ОбходРезультата);
    ВыдатьРекурсивно(ВыборкаДетальная,Номер,ТабДок,Секция,ОбходРезультата);
    КонецЦикла;

ТабДок.ЗакончитьАвтогруппировкуСтрок();

КонецПроцедуры
```


4.2. Конструктор с обработкой результата

В модулях системы в простых случаях (или в целях формирования "каркасного кода") можно использовать возможность вызова "Конструктора с обработкой результата". Данная возможность реализована в виде команды контекстного меню модуля.



Данный конструктор является неким "расширением" конструктора запросов (но в нем отсутствует возможность работы с пакетными запросами).

Проверьте работу этого конструктора на практике.

4.3. Вызов конструктора запроса из тонкого клиента

Начиная с версии платформы 8.3.5, появилась возможность открывать конструктор запроса из тонкого клиента 1С:Предприятие.

Рассмотрим данную возможность с помощью обработки "КонструкторЗапросаУФ".

Текст обработчика команды следующий:

```
&НаКлиенте
```

```
Процедура ОткрытьКонструкторЗапроса(Команда)
```

```
    КЗ=Новый КонструкторЗапроса;
```

```
    ОО = Новый ОписаниеОповещения("ВыполнитьПослеЗакрытияКЗ", ЭтотОбъект);
```

```
    КЗ.Показать(ОО);
```

```
КонецПроцедуры
```

```
Процедура ВыполнитьПослеЗакрытияКЗ(Текст,Параметр)
```

```
ТекДок.ДобавитьСтроку(Текст);
```

```
ОбъектнаяМодельНаСервере(Текст);
```

```
КонецПроцедуры
```

Проверьте работу механизма.

4.4. Использование объектной модели запроса

Начиная с версии платформы 8.3.5, появилась возможность при работе с запросом использовать его объектную модель.

Рассмотрим данную возможность с помощью обработки "КонструкторЗапросаУФ".

Постройте конструктором запрос по справочнику "Номенклатура". С помощью объектной модели запроса добавим в запрос условие на цену продажи.

Текст процедуры следующий:

```
&НаСервере
```

```
Процедура ОбъектнаяМодельНаСервере(Текст)
```

```
ОБ=Новый СхемаЗапроса;
```

```
ОБ.УстановитьТекстЗапроса(Текст);
```

```
ПЗ=ОБ.ПакетЗапросов[0];
```

```
Оператор=ПЗ.Операторы[0];
```

```
Оператор.Отбор.Добавить("ЦенаПродажи = 10");
```

```
ТекДок.Очистить();
```

```
ТекДок.ДобавитьСтроку(ОБ.ПолучитьТекстЗапроса());
```

```
КонецПроцедуры
```

С помощью объектной модели можно не только модифицировать запрос, но и создать его.

```
&НаКлиенте
```

```
Процедура СозданиеЗапросаОбъектнойМоделью(Команда)
```

```
СозданиеЗапросаОбъектнойМодельюНаСервере();
```

```
КонецПроцедуры
```

&НаСервере

Процедура СозданиеЗапросаОбъектнойМодельюНаСервере()

ОБ=Новый СхемаЗапроса;

ПЗ = ОБ.ПакетЗапросов[0];

Оператор = ПЗ.Операторы[0];

Источник =

Оператор.Источники.Добавить(ОБ.ПакетЗапросов[0].ДоступныеТаблицы.Найти("Регистр
Накопления.ОстаткиНоменклатурыСклады.Остатки"));

Источник.Источник.Псевдоним = "Остатки";

Для Каждого ДоступноеПоле Из Источник.Источник.ДоступныеПоля Цикл

ПолеТовар = Оператор.ВыбираемыеПоля.Добавить(ДоступноеПоле);

КонецЦикла;

ВыражениеВыбора = Новый ВыражениеСхемыЗапроса("ВЫБОР

| КОГДА Остатки.КоличествоОстаток > 100

| ТОГДА ""Много""

| ИНАЧЕ ""Мало""

|КОНЕЦ");

ПолеВыбора =

Оператор.ВыбираемыеПоля.Добавить(Строка(ВыражениеВыбора));

Для каждого Колонка Из ПЗ.Колонки Цикл

Если ВРег(Колонка.Поля[0]) = ВРег(ВыражениеВыбора) Тогда

Колонка.Псевдоним = "Оценка";

Прервать;

КонецЕсли;

КонецЦикла;

Для каждого ДоступноеПоле Из Источник.Источник.ДоступныеПоля Цикл

Если ДоступноеПоле.Имя = "Номенклатура" Тогда

ПЗ.Порядок.Добавить(ДоступноеПоле);

```
Прервать;  
КонецЕсли;  
КонецЦикла;  
  
ПЗ.ОбщиеИтоги = Истина;  
ПЗ.КонтрольныеТочкиИтогов.Добавить("Склад");  
  
Для каждого Колонка Из ПЗ.Колонки Цикл  
Если Колонка.Псевдоним = "СуммаОстаток" Тогда  
ПЗ.ВыраженияИтогов.Добавить(Колонка);  
Прервать;  
КонецЕсли;  
КонецЦикла;  
  
Для каждого ВыражениеИтога Из ПЗ.ВыраженияИтогов Цикл  
Если ВыражениеИтога.Поле.Псевдоним = "СуммаОстаток" Тогда  
ВыражениеИтога.Выражение = Новый ВыражениеСхемыЗапроса("Сумма(" +  
ВыражениеИтога.Поле.Псевдоним + ")");  
Прервать;  
КонецЕсли;  
КонецЦикла;  
  
Текст = ОБ.ПолучитьТекстЗапроса();  
ТекДок.Очистить();  
ТекДок.ДобавитьСтроку(Текст);  
  
КонецПроцедуры
```

В итоге получается следующий запрос:

```
ВЫБРАТЬ  
Остатки.Номенклатура КАК Номенклатура,  
Остатки.Склад КАК Склад,  
Остатки.КоличествоОстаток КАК КоличествоОстаток,  
Остатки.СуммаОстаток КАК СуммаОстаток,
```

```
ВЫБОР
      КОГДА Остатки.КоличествоОстаток > 100
          ТОГДА "Много"
          ИНАЧЕ "Мало"
      КОНЕЦ КАК Поле1
ИЗ
      РегистрНакопления.ОстаткиНоменклатурыСклады.Остатки КАК Остатки

УПОРЯДОЧИТЬ ПО
      Остатки.Номенклатура
ИТОГИ
      СУММА(СуммаОстаток)
ПО
      ОБЩИЕ,
      Остатки.Склад
```

Проверьте механизм на практике.

4.5. Работа с вложенными таблицами

Вернемся к понятию "вложенные" таблицы. Рассмотрим отчет "Вложенная таблица". В его модуле расположена следующая процедура:

```
Процедура ВыполнитьОтчет(ТабДок) Экспорт
    Перец ОбщииИтогиСумма;
    Перец ОбщииИтогиКоличество;

    Макет = ПолучитьМакет("ПродолжениеМакет");
    Секция = Макет.ПолучитьОбласть("Шапка");
    ТабДок.Вывести(Секция);
    Секция = Макет.ПолучитьОбласть("Позиция");
    СекцияП = Макет.ПолучитьОбласть("Поставщик");

    Запрос = Новый Запрос;
    ТекстЗапроса = "
```

```
|ВЫБРАТЬ
|Контрагент Как Поставщик,
|Товары.(Номенклатура,Количество)
|ИЗ Документ.ПоступлениеТоваров
|";

Запрос.Текст = ТекстЗапроса;
РезультатЗапроса = Запрос.Выполнить();
ВыборкаОсновная = РезультатЗапроса.Выбрать();
Пока ВыборкаОсновная.Следующий() Цикл
    СекцияП.Параметры.Наименование =
ВыборкаОсновная.Поставщик;
    ТабДок.Вывести(СекцияП);
    ВыборкаДетальная = ВыборкаОсновная.Товары.Выбрать();
    Пока ВыборкаДетальная.Следующий() Цикл
        Секция.Параметры.Наименование =
ВыборкаДетальная.Номенклатура;
        Секция.Параметры.Колво = ВыборкаДетальная.Количество;
        ТабДок.Вывести(Секция);
    КонецЦикла;
КонецЦикла;
КонецПроцедуры
```

Дополните текст запроса следующей строкой и просмотрите полученный результат.

```
Упорядочить По Товары.Номенклатура.Наименование Возр
```

4.6. Использование менеджера временных таблиц

Использование временных таблиц помогает повысить скорость выполнения запросов, сделать процесс построения сложных запросов более простым и организовать исполнение таких запросов "поэтапно".

Возможность использования временных таблиц определяется наличием:

- Объекта "МенеджерВременныхТаблиц"
- Свойства "МенеджерВременныхТаблиц" объекта "Запрос"

- Расширением языка запросов по работе с временными таблицами

В качестве источников данных для временных таблиц могут использоваться:

- Таблица значений
- Табличная часть
- Результат запроса

В тексте запроса для работы с временными таблицами можно использовать следующие ключевые слова:

- Поместить
- Индексировать По
- Уничтожить

Пример работы с временной таблицей:

```
Перем МенеджерВТ;
```

```
Процедура Создание(Кнопка)
```

```
Запрос=Новый Запрос;
```

```
Запрос.МенеджерВременныхТаблиц=МенеджерВТ;
```

```
Запрос.Текст= "ВЫБРАТЬ
```

```
| ОстаткиНоменклатурыОстатки.Номенклатура,
```

```
| ОстаткиНоменклатурыОстатки.КоличествоОстаток,
```

```
| ОстаткиНоменклатурыОстатки.СуммаОстаток
```

```
|ПОМЕСТИТЬ ОстаткиТоваров
```

```
|ИЗ
```

```
| РегистрНакопления.ОстаткиНоменклатуры.Остатки(&ТекущаяДата, ) КАК  
ОстаткиНоменклатурыОстатки";
```

```
Запрос.УстановитьПараметр("ТекущаяДата", ТекущаяДата());
```

```
//Создание временной таблицы
```

```
Результат=Запрос.Выполнить();
```

```
КонецПроцедуры
```

```
Процедура Использование(Кнопка)
```

```
Запрос=Новый Запрос;
```

```
Запрос.МенеджерВременныхТаблиц=МенеджерВТ;
```

```
Запрос.Текст= "ВЫБРАТЬ
```

```
    |      Номенклатура,
```

```
    |      КоличествоОстаток,
```

```
    |      СуммаОстаток
```

```
    |ИЗ
```

```
    |      ОстаткиТоваров";
```

```
Результат=Запрос.Выполнить();
```

```
КонецПроцедуры
```

```
Процедура Удаление(Кнопка)
```

```
    Запрос=Новый Запрос;
```

```
    Запрос.МенеджерВременныхТаблиц=МенеджерВТ;
```

```
    Запрос.Текст= " Уничтожить ОстаткиТоваров";
```

```
    Результат=Запрос.Выполнить();
```

```
КонецПроцедуры
```

```
Процедура ТЗ(Кнопка)
```

```
    Запрос=Новый Запрос;
```

```
    // В свойство запроса "МенеджерВременныхТаблиц" записываем менеджер,
```

```
    // в контексте которого создается временная таблица
```

```
    Запрос.МенеджерВременныхТаблиц=МенеджерВТ;
```

```
    Запрос.Текст= " ВЫБРАТЬ
```

```
        |      Товар,
```

```
        |      Количество
```

```
        |ПОМЕСТИТЬ ТЧРеализацииИОстатки
```

```
        |ИЗ
```

```
        |      &ВнешнийИсточник Как Внешний ";
```

```
    Запрос.УстановитьПараметр("ВнешнийИсточник",ЭлементыФормы.ТПРезультатЗапроса.Значение);
```

```
    Результат=Запрос.Выполнить();
```


ЭлементыФормы.ТПРезультатЗапроса.Значение=Результат.Выгрузить();

ЭлементыФормы.ТПРезультатЗапроса.СоздатьКолонки();

КонецПроцедуры

МенеджерВТ=Новый МенеджерВременныхТаблиц;

Важно!

При использовании в качестве источника данных временной таблицы, колонка таблицы не может иметь тип "Характеристика".

При использовании в качестве источника данных для временной таблицы "внешней" таблицы значений, колонки таблицы значения должны быть явно типизированы.

При создании временной таблицы в процессе исполнения запроса, в нем нельзя использовать раздел "Итоги".

Практикум № 24

Откорректируйте в соответствии с новыми возможностями запрос в обработке проведения документа "РеализацияТоваров".

4.7. Использование предопределенных данных конфигурации

В тексте запроса можно "напрямую" (без использования параметров запроса) определять ссылки на предопределенные данные информационной базы, значения некоторых системных перечислений, пустые ссылки, значения точек маршрута карты бизнес-процессов.

Обращение реализовано с помощью литерала "Значение(Описание значения)".

Для предопределенных данных используется следующий способ обращения:

Значение(ВидОбъекта.ИмяОбъектаКонфигурации.Значение).

В качестве вида объекта может выступать:

- Справочник
- ПланВидовХарактеристик
- ПланСчетов
- ПланВидовРасчета
- Перечисление

Для обращения к системным перечислениям используется следующий вариант:

Значение(ИмяСистемногоПеречисления.Значение)

Можно обращаться к следующим системным перечислениям:

- ВидДвиженияНакопления
- ВидДвиженияБухгалтерии
- ВидСчета

Для получения пустой ссылки используется:

Значение(ВидОбъекта.ИмяОбъектаКонфигурации.ПустаяСсылка)

Для обращения к точкам маршрута бизнес-процесса:

Значение(БизнесПроцесс.ИмяБизнесПроцесса.ТочкаМаршрута.Значение)

Пример обращения к предопределенным данным:

```
Запрос=Новый Запрос;
Запрос.Текст="ВЫБРАТЬ
|     ЦеныНоменклатуры.Номенклатура,
|     ЦеныНоменклатуры.Цена
|ИЗ
|     РегистрСведений.ЦеныНоменклатуры КАК ЦеныНоменклатуры
|ГДЕ
|     ЦеныНоменклатуры.ТипЦен = ЗНАЧЕНИЕ(Справочник.ТипыЦен.Основной)";
Результат=Запрос.Выполнить();
ЭлементыФормы.ТПРезультатЗапроса.Значение=Результат.Выгрузить();
ЭлементыФормы.ТПРезультатЗапроса.СоздатьКолонки();
```

Практикум № 25

Получите элементы справочника "Номенклатура" у которых не заполнена основная единица измерения.

4.8. Возможности конструкции "В"

Сравните результат исполнения двух запросов

Запрос №1

ВЫБРАТЬ

ОстаткиНоменклатурыСкладыОстатки.Номенклатура,
ОстаткиНоменклатурыСкладыОстатки.Склад,
ОстаткиНоменклатурыСкладыОстатки.КоличествоОстаток,
ОстаткиНоменклатурыСкладыОстатки.СуммаОстаток

ИЗ

РегистрНакопления.ОстаткиНоменклатурыСклады.Остатки(

Номенклатура В

(ВЫБРАТЬ

ПоступлениеТоваровСкладыТовары.Номенклатура

ИЗ

Документ.ПоступлениеТоваровСклады.Товары КАК
ПоступлениеТоваровСкладыТовары

ГДЕ

ПоступлениеТоваровСкладыТовары.Ссылка
= &ЗначениеСсылка)

И Склад В

(ВЫБРАТЬ

ПоступлениеТоваровСкладыТовары.Склад

ИЗ

Документ.ПоступлениеТоваровСклады.Товары КАК
ПоступлениеТоваровСкладыТовары

ГДЕ

ПоступлениеТоваровСкладыТовары.Ссылка
= &ЗначениеСсылка)) КАК ОстаткиНоменклатурыСкладыОстатки

Запрос №2

```
ВЫБРАТЬ
    ОстаткиНоменклатурыСкладыОстатки.Номенклатура,
    ОстаткиНоменклатурыСкладыОстатки.Склад,
    ОстаткиНоменклатурыСкладыОстатки.КоличествоОстаток,
    ОстаткиНоменклатурыСкладыОстатки.СуммаОстаток
ИЗ
    РегистрНакопления.ОстаткиНоменклатурыСклады.Остатки(
        ,
        (Номенклатура, Склад) В
            (ВЫБРАТЬ
                ПоступлениеТоваровСкладыТовары.Номенклатура,
                ПоступлениеТоваровСкладыТовары.Склад
            ИЗ
                Документ.ПоступлениеТоваровСклады.Товары КАК
ПоступлениеТоваровСкладыТовары
            ГДЕ
                ПоступлениеТоваровСкладыТовары.Ссылка =
&ЗначениеСсылка)) КАК ОстаткиНоменклатурыСкладыОстатки
```

4.9. Получение всех результатов пакетного запроса

Как было сказано ранее, пакетный запрос это возможность в тексте указать сразу несколько текстов запроса (разделяя их ";"). При использовании метода "Выполнить" объекта "Запрос" будет получен результат последнего запроса. Т.е. исполнение начинается с самого первого. В первых обычно создаются временные таблицы, в последующих запросах они используются. Но существует возможность получить результаты всех запросов, входящих в пакет (это удобно, когда в пакет включены "не зависимые запросы"). Пример кода приводится ниже:

```
Результат=Запрос.ВыполнитьПакет();

Для Каждого ЭлементРезультата Из Результат Цикл

    ТП=ЭлементРезультата.Выгрузить();

    ЭлементыФормы.ТП.СоздатьКолонки();

    Предупреждение("Пауза");

КонецЦикла;
```

Практикум № 26

1. Получите предыдущий курс указанной валюты (предшествующий актуальному).
 2. Получите предыдущие курсы всех валют (когда-либо внесенных в регистр сведений).
 3. Получите данные о номенклатуре, текущему остатку этой номенклатуры, по которой эти остатки изменялись (соответствующая информация должна браться из планов обмена).
 4. Дополнительно к данным, получаемым по пункту 4, в результате запроса должны присутствовать данные о документе сделавшем изменение остатка и величине изменения.
 5. Напишите текст запроса, который выдавал бы список покупателей, покупавших не более двух позиций в каждом документе (во всех документах было не более 2х строк в табличной части). При этом нельзя (с точки зрения усложнения задачи) использовать конструкцию "Имеющие"
-

5. Типовые ошибки

Рассматривая "различные произведения, различных авторов" можно выделить ряд наиболее часто встречающихся ошибок (синтаксические ошибки не в счет):

- Попытка упорядочить по ссылочным полям (без использования ключевого слова "Автоупорядочивание"), неправильное использование варианта упорядочивания "По иерархии".
- Не использование параметров виртуальных таблиц (описание условий, дат в разделе "Где", а не в параметрах виртуальных таблиц)
- Неверное использование раздела "Имеющие" (определяются условия, которые должны определяться либо в параметрах виртуальных таблиц, или разделе "Где").
- Неверное использование таблиц (попытка получения оборотов из таблицы остатков и оборотов и т.п., попытка реализовать функциональность виртуальных таблиц на реальных таблицах...).

Отдельно можно выделить ряд пожеланий, следование которым поможет строить более оптимальные запросы:

- Не желательно упорядочивать по виртуальным полям таблиц (например, по представлению).
- Если результат запроса будет выводиться в печатную форму рекомендуется в поля запроса включать представления (использовать соответствующую функцию языка запросов) или наименования.

6. Рекомендация "напоследок"

Читайте соответствующие разделы на дисках "ИТС". Там можно почерпнуть массу полезной (в том числе и по запросам) информации. К примеру: рекомендации по использованию параметров виртуальных таблиц, отборов при написании текста запроса и т.д.

